
基于球极投影的双曲几何结构可视化研究

龚佑康

（数学系 指导教师：朱一飞）

[摘要] 双曲几何作为非欧几何的重要分支，其常负曲率特性与欧氏几何

的直觉相悖，使得抽象度量结构的直观理解存在困难。本文系统推导了球

极投影的正逆映射公式，严格证明了其保角性与保圆性；并梳理了双曲几

何的经典模型体系与正则密铺理论。本文旨在利用球极投影这一共形映射

工具，构建双曲几何经典模型及其在前沿物理应用中的可视化框架，并借

助三维打印技术制作实体展示模型。

[关键词]：球极投影; 双曲几何; 可视化模型



[ABSTRACT]: As an important branch of non-Euclidean geometry,

hyperbolic geometry is characterized by a constant negative curvature that

defies Euclidean intuition, posing significant challenges to the intuitive

comprehension of its abstract metric structure. This thesis systematically derives

the forward and inverse mapping formulas for stereographic projection and

rigorously proves its angle-preserving and circle-preserving properties.

Furthermore, it comprehensively reviews the classical model representations and

regular tessellation theory of hyperbolic geometry. By using stereographic

projection as a conformal mapping tool, this work establishes a visualization

framework for classical hyperbolic models and their applications in frontier

physics, also provides physical display models with 3D printing technology.

[Key words]:Hyperbolic Geometry; Stereographic Projection ; Visual Model

1. 简介

自欧几里得《几何原本》确立公理体系以来，平行公设始终作为构建平坦空间观念

的逻辑基石。直至 19 世纪，罗巴切夫斯基、波尔约与高斯等人各自通过对第五公设的

反思，独立创立了非欧几何，从而引发了空间观念的变革。作为非欧几何的核心分支，

双曲几何以“过直线外一点至少有两条直线与已知直线平行”替代平行公设，双曲空间

的曲率变为负常数，在此恒负曲率的空间中，几何对象呈现出异于欧氏几何的内在特性：

三角形内角和严格小于平角，其面积由角亏决定。全新的拓扑与度量性质不仅重构了人

类对几何基础的认知，亦在诸多领域有所发挥。

然而，受限于欧几里得空间的直觉感知，双曲几何的抽象度量结构为其直观理解与

科学传播带来了显著障碍。如何在欧氏空间中精确表征双曲空间，成为数学家长期探索

的焦点。为此，数学界发展了多种等距或保角模型，如克莱因圆盘模型、庞加莱圆盘模

型、上半平面模型及双曲面模型。在这些不同模型间的相互转换与映射过程中，球极投

影作为一种经典的共形映射工具，发挥了重要的桥梁作用。球极投影在映射过程中严格

保持角度的不变性（共形性）与广义圆的对应关系（保圆性），这种几何不变量保持特

性，提供了优秀的可视化条件。

近年来，双曲几何的理论框架已突破纯粹数学的边界，深度渗透至现代凝聚态物理

与量子模拟领域，兴起了双曲物理的交叉研究前沿。在传统的欧几里得晶格体系中，离

散平移对称性所对应的阿贝尔群结构确保了布洛赫定理的适用性，构成了经典能带理论

的基础。然而在负曲率空间中，平移操作呈现非交换性，导致了传统理论的失效，进而

衍生出高维布里渊区与双曲能带拓扑等全新物理图景。

本文旨在利用球极投影的共形映射机制，构建双曲几何经典模型及其在前沿物理应

用中的可视化框架，并借助三维打印技术展示。

在第二节中，本文将从三维直角坐标系出发，严格推导球极投影的正、逆映射解析

公式并引入复表示，利用代数方程与雅可比矩阵严密论证其保圆性与保角性，为后续可

视化工作提供坚实的理论支撑。

在第三节中，本文还将梳理双曲几何的基础概念与模型体系，对比分析庞加莱圆盘、

上半平面、克莱因圆盘及双曲面模型中的度量与距离，并探讨双曲密铺理论，展示负曲

率空间中正则密铺的无限多样性。

在第四节中，本文将聚焦于两份前沿研究中涉及的双曲模型的可视化实践。其一，

基于 Qiaolu Chen等人的研究
[1]
，构建将 {8,3} 双曲晶格投影至庞加莱圆盘的三维打印模

型；其二，基于 Alicia J. Kollár等人的研究
[2]
，制作投影至圆盘的 {7,3} 双曲镶嵌及其衍

生有效笼目（kagome）晶格的实体模型。

2.球极投影的解析公式与几何性质

本节主要以最常见的二维球面为例，给出相应的球极投影公式推导与性质分析。高

维情形的公式可参考
[3]
。

2.1 解析公式推导

建立三维直角坐标系，设球面的方程为 �2 + �2 + �2 = �2 ，北极点 � 的坐标为

(0,0, �) 。取赤道平面 � = 0 作为投影平面。

设球面上任意一点 � 的坐标为 (�, �, �) (� ≠ �) ，其在投影平面上的对应点 �' 的坐

标为 (�, �, 0) 。由于 �、�、�' 三点共线，向量 ������� 与 ��'������ 平行，故存在实数 � 使得：

��'������ = �������� (1)

代入三点坐标有：

(�, �, − �) = �(�, �, � − �) (2)

对应求解即得球极投影的正映射公式：

� =
�

� − �
，� = �� =

��
� − �

，� = �� =
��

� − �
(3)

反之，给定投影平面上的点 �'(�, �, 0) ，将 (3) 式中的对应关系代入球面方程：

� =
�
� ，� =

�
� ，� = � −

�
�

(4)

�2

�2 +
�2

�2 + � −
�
�

2

= �2 (5)

整理得到：

�2 + �2 + �2

�2 =
2�2

�
(6)

� =
�2 + �2 + �2

2�2 (7)

进而求得逆映射公式：

� =
2�2�

�2 + �2 + �2 ，� =
2�2�

�2 + �2 + �2 ，� =
�(�2 + �2 − �2)

�2 + �2 + �2 (8)

对于不同的投影平面仅需在 (2) 式稍作修改后求解，这里不再赘述。

注意到该映射为双射，且由 � − � > 0，�2 + �2 + �2 > 0 知正、逆映射均为连续映

射，则去除了北极点的球面 �2 ∖ � 与二维欧氏平面 ℝ2 是同胚的。在此基础上，我们

引入无穷远点，使得扩充平面 ℝ�2 = ℝ2 ∪ ∞ ，规定北极点 � 与无穷远点 ∞ 相对应，

并且当球面上的点 � 趋近于北极点 � 时，其投影点 �' 趋向于 ∞ ；反之亦然。此时容易

发现球面 �2 与扩充平面 ℝ�2 同胚。

当我们将二维平面 ℝ2 看作复平面 ℂ 时，此时扩充平面 ℝ�2 变为扩充复平面 ℂ� = ℂ ∪

∞ ，也被称为黎曼球面。相关内容可参考
[4]
。

令 � = � + �� ，引入复数后，正向映射公式 (3) 可改写为复数形式：

� = � + �� =
�(� + ��)

� − �
(9)

逆映射公式 (8) 变为：

� =
�( � 2 − �2)

� 2 + �2 ，� + �� =
�(� − �)

�
=

�
�

� −
�( � 2 − �2)

� 2 + �2 =
2�2�

� 2 + �2 (10)

2.2 几何性质分析

我们将分析与证明球极投影的两个核心几何性质：保圆性和保角性。

保圆性指球极投影将球面上的任意圆周映射为平面上的广义圆，反之亦然。平面上

的广义圆可以用统一的方程表示：

�(�2 + �2) + �� + �� + � = 0 (11)

其中，当 � = 0 时，该方程退化为直线。将公式 (3) 代入得到：

�
�2(�2 + �2)

(� − �)2 + �
��

� − �
+ �

��
� − �

+ � = 0 (12)

注意到 �2 + �2 = �2 − �2 = (� − �)(� + �) ，代入并化简得：

��2(� − �)(� + �) + (��� + ���)(� − �) + �(� − �)2 = 0 (13)

又因为 � − � ≠ 0 ，两边约去并整理得：

(��)� + (��)� + (��2 − �)� + (��3 + ��) = 0 (14)

注意到 (14) 是关于 �，�，� 的一次方程，表示空间中的一个平面，又因为点

(�, �, �) 在球面上，故该点的集合是此平面与球面的交线，即球面上的一个圆周。特别

地，当 � = 0 时（平面上为直线）， (14) 变为 (��)� + (��)� − �� + �� = 0 ，该平面

恒过北极点 (0,0, R) ，因此对应球面上过北极点的圆周。类似可证明球面上的圆周映射

为平面上的广义圆。

保角性又称共形性，指球极投影在映射前后保持任意两条曲线的交角不变，即球面

上两条相交曲线的切线夹角，等于它们在平面上投影曲线的切线夹角。

考察从平面到球面的映射，计算该映射的雅可比矩阵（Jacobian Matrix），若该映

射在任意点的雅可比矩阵满足其转置与自身乘积正比于单位矩阵，则说明该映射是一个

相似变换，从而保持向量间的夹角不变。

对逆映射公式 (8) 计算 �，�，� 关于 �，� 的偏导，可得雅可比矩阵：

� =
2�2

�2 + �2 + �2 2

�2 + �2 − �2 −2��
−2�� �2 + �2 − �2

2�� 2��
(15)

��� =
4�4

(�2 + �2 + �2)2
1 0
0 1 (16)

设平面上两条相交曲线在交点处的切向量分别为 � 和 � ，映射到球面后的切向量分

别为 �� 和 �� 。它们在映射前后的夹角余弦值分别为：

cos � =
� ⋅ �
|�||�|， cos �' =

(��) ⋅ (��)
|��||��| =

��(���)�

��(���)� ��(���)�
(17)

将 (16) 式代入得：

cos �' =
�2(� ⋅ �)

�2|�|2 �2|�|2
=

�2(� ⋅ �)
�2|�||�|

=
� ⋅ �
|�||�|

= cos � (18)

由此得到 �' = � ，映射前后夹角相等，则球极投影的保角性得证。

2.3 投影模型展示

将球极投影的数学概念进行可视化与实体化展示，是加深理解的重要途径。基于球

极投影的保圆性与保角性，其生成的几何图案兼具严密的逻辑性与独特的美学价值，本

文完成了一份投影平面网格的三维打印模型的制作，如图 1展示。

图 1 网格投影模型

在应用层面，球极投影广泛存在于科学与工程的诸多领域中。在制图学中，保角性

使得球极投影成为绘制极地航海图的首选，尽管面积会产生边缘畸变，但航向角度的精

准保持确保了航海路线的直线对应；在结晶学与地质学中，晶面法线和岩层产状的球面

数据通过球极投影转化为平面网格图，使得空间角度的测算转化为平面作图问题，极大

地简化了分析流程；在计算机图形学中，球极投影常被用于环境贴图和全景图的映射，

其无缝且保角的特性有效避免了图像在拼接处的视觉撕裂。

3.双曲几何的模型体系与密铺分类

双曲几何作为非欧几何的典型代表，不仅从根本上改变了人类对空间与公理体系的

认知，更为现代数学、理论物理以及复杂网络的可视化提供了核心理论框架。在探讨通

过球极投影在平面上展示双曲几何之前，本节将梳理双曲几何的基础概念、经典模型表

示方法以及基于双曲空间的密铺理论。

3.1 基本概念与性质

双曲几何的公设体系在保留欧几里得几何前四条公设的基础上，将第五公设替换为：

过直线外一点，至少存在两条直线与已知直线不相交。这一修改导致了空间曲率的根本

性改变，双曲空间成为具有常负曲率（通常取 � = − 1 ）的度量空间。

在曲率为 − 1 的双曲空间中，三角形内角和 � + � + � < � ，其差值 � = � − (� +

� + �) 称为角亏。角亏直接决定了双曲三角形的面积，即 ���� = � ，可由 Gauss-Bonnet

定理直接导出。这意味着在双曲空间中，不存在任意大的三角形，且相似即全等。

3.2 经典模型分类

为了在欧氏空间中直观表示双曲几何，数学家构建了多种的模型。常用的双曲几何

模型可分为以下四类：

1.庞加莱圆盘模型

庞加莱圆盘模型是本文关注的核心。它将整个双曲平面映射到欧氏平面的单位圆盘

� = {� ∈ ℂ: |�| < 1}内。双曲空间中的点对应圆盘内的点；双曲空间中的测地线对应于

与 �� 正交的圆弧或直径。庞加莱圆盘模型是共形的，即模型中两曲线相交的欧氏角度

等于双曲空间中的真实角度。其度量形式为：

��2 =
4(��2 + ��2)
(1 − �2 − �2)2 (19)

对任意两点 �1，�2 ∈ � ，其双曲距离函数为：

��(�1, �2) = ������ℎ 1 +
2|�1 − �2|2

(1 − |�1|2)(1 − |�2|2)
(20)

该距离公式表明，当点趋近于圆盘边界时，距离趋于无穷大。模型中越靠近边缘的物体，

其在欧氏视觉上越小，单位欧氏线段所代表的双曲长度呈指数级增长。

2.庞加莱上半平面模型

该模型将双曲空间映射到复平面的上半部分 ℍ = {� ∈ ℂ: ��(�) > 0} ，测地线为垂

直于实轴的半圆，或垂直于实轴的竖直直线。其度量形式为：

��2 =
��2 + ��2

�2 (21)

对任意两点 �1 = �1 + ��1，�2 = �2 + ��2 ∈ ℍ ，其双曲距离函数为：

�ℍ(�1, �2) = ������ℎ 1 +
|�1 − �2|2

2�1�2
(22)

上半平面模型与圆盘模型通过莫比乌斯变换建立等距同构，映射 �(�) = ��+1
�+�

将上半平面

映射到单位圆盘内部。

3.克莱因圆盘模型

克莱因模型同样使用单位圆盘 � = {(�, �) ∈ ℝ2: �2 + �2 < 1} ，但测地线是欧氏弦

（即圆盘内的直线段）。其度量形式为：

��2 =
(1 − �2)��2 + (1 − �2)��2 + 2�� �� ��

(1 − �2 − �2)2 (23)

对任意两点 �，� ∈ � ，其双曲距离函数为：

��(�, �) = ������ℎ
1 − � ⋅ �

(1 − ‖�‖2)(1 − ‖�‖2)
(24)

克莱因模型不是共形的，度量中包含交叉项，表明在不同方向上的缩放是不一致的，角

度会发生畸变。

4.双曲面模型

双曲面模型将双曲空间 ℍ2 等距嵌入三维闵可夫斯基空间 ℝ2,1 中，由双曲面的向前

叶 �+ = {[�0, �1, �2]: �02 − �12 − �22 = 1, �0 > 0} 构成。其度量形式为：

��2 =− ��02 + ��12 + ��22 (25)

任意两点 � = �0, �1, �2 ，� = �0, �1, �2 间的双曲距离函数为：

�(�, �) = ������ℎ �0�0 − �1�1 − �2�2 (26)

关于各个模型相关的计算与性质可以对照
[3-5]

中对应模型的章节。

3.3 双曲密铺理论

密铺是几何学中的经典问题，指利用特定的几何图形无重叠且无间隙地覆盖整个平

面。在欧几里得几何中，受限于正多边形内角和必须恰好拼合成 360度的硬性约束，仅

存在三种正则密铺，即由正三角形（ {3,6} ）、正方形（ {4,4} ）和正六边形（ {6,3} ）

构成的密铺。而在双曲几何中，由于空间具有常负曲率，多边形的内角和严格小于同等

边数的欧氏多边形，且随着多边形面积的增大，其内角和会逐渐趋近于零。这一由“角

亏”决定的独特性质，打破了欧氏空间的严苛限制，使得双曲空间中存在无穷多种正则

密铺。

在双曲平面上，正则密铺通常用 Schläfli 符号 {�, �} 来表示，其中 � 代表正多边形

的边数，� 代表每个顶点处相邻的多边形个数。根据双曲空间的几何性质，双曲正则密

铺存在的充要条件为 (� − 2)(� − 2) > 4 。只要满足此不等式，如 {5,4} （四个正五边

形交汇于一点）、 {7,3} （三个正七边形交汇于一点）乃至边数与顶点数更多的组合，

均能在双曲平面上实现无隙无叠的完美铺满。这种种类上的无限丰富性，是欧氏几何与

球面几何所不具备的。

在可视化表示方面，双曲密铺通常借助庞加莱圆盘模型进行展示。由于该模型中越

靠近圆盘边界，单位欧氏线段所代表的双曲长度呈指数级增长，因此在视觉呈现上，处

于圆盘中心的多边形显得较大且形状规则，而向边缘延伸的多边形则在欧氏视觉上急剧

缩小，呈指数级密集地排列并收敛于边界。这种视觉上的指数级收敛，恰好直观地再现

了双曲空间无限广阔且面积呈指数级增长的度量本质。Visualizing Mathematics with 3D

Printing[6]
中展示了诸多优美的双曲密铺图案。

双曲密铺理论不仅是非欧几何的直观图解，在数学基础研究中占有重要地位，更在

众多前沿领域展现出深刻的应用价值。在几何群论中，它为研究无限群的结构提供了直

观的 Cayley 图表示；在艺术领域，著名版画家 M.C.埃舍尔的《圆周极限》系列正是基于

庞加莱圆盘上的双曲密铺创作而成；在现代计算科学中，双曲密铺天然的层级与树状几

何特性，被广泛用于复杂网络的可视化布局、层级数据的结构化嵌入以及双曲图神经网

络的设计，为处理具有隐式树状结构的大规模数据提供了极其重要的几何框架。

4.相关研究的模型可视化展示

本节将聚焦于两篇研究中构建的非欧几何模型，通过球极投影这一共形映射工具可

视化这些模型。需要说明的是，本节展示的模型仅实现了数学结构的可视化——球极投

影将双曲空间中的晶格映射到庞加莱圆盘或双曲面上。原论文中那些精彩的物理性质

（拓扑边缘态的单向传输、Chern数、反常 Floquet相等），模型本身无法直接体现。但

正如一块芯片的物理性能源自其内部纳米级的电路布局，这些双曲晶格的非平凡拓扑性

质，根植于本节模型所展示的几何结构之中。也因此，可视化的几何结构并非与物理无

关——它是所有后续物理讨论的平台。

4.1 非互易散射网络中的双曲晶格

本文基于论文 Anomalous and Chern topological waves in hyperbolic networks[1]中涉

及的 {8,3} 双曲晶格，制作了从球面投影到庞加莱圆盘与双曲面上的可视化三维打印模

型（图）。近年来，拓扑物态的研究已从欧几里得空间拓展至非欧几里得空间。2024

年，Chen等人在 Nature Communications上发表的研究在实验上实现了非互易双曲散射

网络中的拓扑手性边缘模传输 。该工作利用三端口非互易酉散射体构建 {8,3} 双曲晶格，

通过直接场映射实验证明拓扑边缘模态的单向特性，可视化异常手性边缘模沿边界顺时

针传输。图 2 模型呈现了庞加莱圆盘上的三层八边形密铺，图中平面上标注了 {8,8} 双

曲布拉维晶格的中心单元格，研究中在这个标注八边形中包含的 16 个顶点处布置了三

端口非互易环形器作为一整个单元格分析。

图 2 {�, �} 晶格图及黑线标注 {�, �} 中心格

4.2 电路 QED中的双曲晶格

本文基于论文 Hyperbolic lattices in circuit quantum electrodynamics[2]中涉及的 {7,3}

布局晶格及其产生的有效笼目（kagome）晶格，制作了投影至圆盘的实体模型（图）。

近年来，量子模拟的研究也已从欧几里得晶格拓展至弯曲空间中的晶格体系。2019年，

Kollár等人在 Nature 上发表的研究在电路量子电动力学平台上关联了双曲晶格 。该工

作利用超导共面波导谐振器的可变形特性，将原本无法在欧氏空间中无畸变嵌入的双曲

平面纳入实验对象，通过传输测量观测到了与理论吻合的能谱特征，为量子模拟与强关

联物理研究开辟了全新维度 。图 3模型展示了研究中的 {7,3} 布局晶格，谐振器位于七

边形的顶点，图中还标注了局部的连接七边形各边中点所得的有效笼目晶格。

图 3 {�, �} 晶格及局部的有效笼目晶格

结束语：

本文系统梳理了球极投影的解析推导与几何性质，构建了从球面到欧氏平面的严格

映射框架；在此基础上，对比分析了双曲几何的四大经典模型与正则密铺理论，揭示了

负曲率空间中度量膨胀与拓扑多样性的内在联系。通过引入前沿物理中的双曲晶格研究，

本文利用球极投影的共形性质完成了从抽象数学结构到三维实体模型的可视化。

制作的模型在展示内容与意义等方面依然存在局限性，未来仍可进一步探索如何以

更具关联性的可视化方式将物理性质与双曲模型结合；以及拓展至更高维双曲空间的投

影可视化与三维打印工艺优化。非欧几何模型的实体化与交互式展示，将始终为数学教

育、物理建模与研究提供直观的几何语言。

附录

A. Mathematica代码生成可投影出双曲密铺图案的球壳模型

ClearAll["Global`*"]

(*1.密铺参数*)

p = 7; q = 3; tilingDepth = 3;

(*2.双曲几何基础*)

h2eNorm[h_] := N[Tanh[h/2]];

halfEdgeLength = N[ArcCosh[Cos[Pi/p]/Sin[Pi/q]]];

circumRadius = N[ArcSinh[Sinh[halfEdgeLength]/Sin[Pi/p]]];

rVertex = h2eNorm[circumRadius];

HyperbolicTranslate[P_?NumericQ] :=

Function[z, (z + P)/(Conjugate[P] z + 1)];

HyperbolicRotate[theta_?NumericQ] := Function[z, Exp[I theta] z];

ReflectAcrossGeodesic[z0_?NumericQ, z1_?NumericQ] :=

Module[{tToOrigin, z1AtOrigin, angle, rotToX, rotFromX},

tToOrigin = HyperbolicTranslate[-z0];

z1AtOrigin = tToOrigin[z1];

angle = Arg[z1AtOrigin];

rotToX = HyperbolicRotate[-angle];

rotFromX = HyperbolicRotate[angle];

Function[z, Module[{w}, w = tToOrigin[N[z]]; w = rotToX[w];

w = Conjugate[w]; w = rotFromX[w];

N[HyperbolicTranslate[z0][w]]]]];

HyperbolicLerp[p0_, p1_, t_] :=

Module[{translateToOrigin, p1Prime, normSq, norm, atanNorm,

desiredNorm, direction},

translateToOrigin = HyperbolicTranslate[-N[p0]];

p1Prime = translateToOrigin[N[p1]]; normSq = Abs[p1Prime]^2;

If[normSq < 10^-12, Return[N[p0 + t (p1 - p0)]]];

If[normSq > 1.001, Return[Undefined]];

If[normSq > 1, normSq = 1.0];

norm = Sqrt[normSq]; atanNorm =ArcTanh[norm];

desiredNorm = Tanh[t*atanNorm];

direction = If[norm > 10^-12, p1Prime/norm, 0];

N[HyperbolicTranslate[N[p0]][direction*desiredNorm]]];

SubdivideHyperbolicEdge[p0_, p1_, nSeg_ : 30] :=

Table[HyperbolicLerp[p0, p1, t], {t, 0., 1., 1./nSeg}];

(*3.BFS 生成密铺*)

centerVerts = Table[N[rVertex Exp[I 2 Pi k/p]], {k, 0, p - 1}];

GenerateTiling[maxDepth_, maxPolys_ : 400] :=

Module[{polys = {}, seen = <||>, queue = {}, tol = 10^-4, count = 0,

key}, AppendTo[polys, centerVerts];

AssociateTo[seen, Round[N[Mean[centerVerts]], tol] -> True];

queue = {{centerVerts, 0}}; count = 1;

While[Length[queue] > 0 && count < maxPolys,

Module[{verts, depth, z0, z1, reflect, newVerts, newCenter,

k}, {verts, depth} = First[queue]; queue = Rest[queue];

If[depth >= maxDepth, Continue[]];

Do[z0 = verts[[k]]; z1 = verts[[Mod[k, p] + 1]];

reflect = ReflectAcrossGeodesic[z0, z1];

newVerts = reflect /@ verts; newCenter = Mean[newVerts];

If[Abs[newCenter] > 0.998, Continue[]];

key = Round[newCenter, tol];

If[! KeyExistsQ[seen, key], AppendTo[polys, newVerts];

AssociateTo[seen, key -> True];

AppendTo[queue, {newVerts, depth + 1}]; count++], {k, 1, p}]]];

polys];

tilingPolys = GenerateTiling[tilingDepth];

(*4.提取唯一边*)

ExtractUniqueEdges[polys_] :=

Module[{edgeMap = <||>, tol = 10^-4, edges = {}, v1, v2, key},

Do[Do[v1 = Round[N[poly[[k]]], tol];

v2 = Round[N[poly[[Mod[k, Length[poly]] + 1]]], tol];

key = If[Abs[v1] <= Abs[v2], {v1, v2}, {v2, v1}];

If[! KeyExistsQ[edgeMap, key], AssociateTo[edgeMap, key -> True];

AppendTo[

edges, {poly[[k]], poly[[Mod[k, Length[poly]] + 1]]}]], {k,

Length[poly]}], {poly, polys}];

edges];

allEdges = ExtractUniqueEdges[tilingPolys];

(*5.逆映射到南半球*)

InvStereoSouth[{x_?NumericQ, y_?NumericQ}] :=

Module[{r2 = x^2 + y^2, d}, d = 1 + r2; {2 x/d, 2 y/d, (r2 - 1)/d}];

SafeStereoSouth[z_?NumericQ, R_?NumericQ] :=

Module[{x, y, r, scale}, x = Re[N[z]]; y = Im[N[z]];

r = Sqrt[x^2 + y^2];

If[r > 0.999, scale = 0.999/r; x *= scale; y *= scale];

R*InvStereoSouth[{x, y}]];

(*6.实体棱柱生成*)

GenerateSolidPrismsSouth[edges_, width_, nSeg_, R_, h_] :=

Module[{allVerts = {}, allTris = {}, vCount = 0, upper2D, lower2D,

m, uo, ui, lo, li, base, i, edge, p0, p1, pts, dirs, perps},

Do[p0 = edge[[1]]; p1 = edge[[2]];

pts = SubdivideHyperbolicEdge[p0, p1, nSeg]; m = Length[pts];

dirs =

Table[With[{diff = pts[[Min[i + 1, m]]] - pts[[Max[i - 1, 1]]]},

If[Abs[diff] > 10^-12, diff/Abs[diff], I]], {i, m}];

perps = Table[{-Im[dirs[[i]]], Re[dirs[[i]]]}, {i, m}];

perps =

Table[If[Norm[perps[[i]]] > 10^-12,

perps[[i]]/Norm[perps[[i]]], {0, 1}], {i, m}];

upper2D = Table[pts[[i]] + (width/2)*Complex @@ perps[[i]], {i, m}];

lower2D = Table[pts[[i]] - (width/2)*Complex @@ perps[[i]], {i, m}];

uo = SafeStereoSouth[#, R] & /@ upper2D;

ui = SafeStereoSouth[#, R - h] & /@ upper2D;

lo = SafeStereoSouth[#, R] & /@ lower2D;

li = SafeStereoSouth[#, R - h] & /@ lower2D;

For[i = 1, i < m, i++, base = vCount;

allVerts =

Join[allVerts, {uo[[i]], uo[[i + 1]], lo[[i + 1]], lo[[i]],

ui[[i]], ui[[i + 1]], li[[i + 1]], li[[i]]}];

allTris =

Join[allTris, {{base + 1, base + 2, base + 3}, {base + 1,

base + 3, base + 4}, {base + 5, base + 8,

base + 7}, {base + 5, base + 7, base + 6}, {base + 1,

base + 5, base + 6}, {base + 1, base + 6,

base + 2}, {base + 4, base + 3, base + 7}, {base + 4,

base + 7, base + 8}}];

vCount += 8;], {edge, edges}];

{allVerts, allTris}];

(*7.自适应碗口补齐*)

GenerateBowlRim[zBottom_, zTop_, R_, h_, nSeg_ : 100] :=

Module[{rimVerts = {}, rimTris = {}, theta, rOutB, rInB, rOutT,

rInT, base, i}, rOutB = Sqrt[Max[0, R^2 - zBottom^2]];

rInB = Sqrt[Max[0, (R - h)^2 - zBottom^2]];

rOutT = Sqrt[Max[0, R^2 - zTop^2]];

rInT = Sqrt[Max[0, (R - h)^2 - zTop^2]];

For[i = 0, i <= nSeg, i++, theta = 2 Pi*i/nSeg;

AppendTo[rimVerts, {rOutB Cos[theta], rOutB Sin[theta], zBottom}];

AppendTo[rimVerts, {rInB Cos[theta], rInB Sin[theta], zBottom}];

AppendTo[rimVerts, {rOutT Cos[theta], rOutT Sin[theta], zTop}];

AppendTo[rimVerts, {rInT Cos[theta], rInT Sin[theta], zTop}]];

For[i = 0, i < nSeg, i++, base = 4*i;

AppendTo[rimTris, {base + 1, base + 5, base + 3}];

AppendTo[rimTris, {base + 5, base + 7, base + 3}];

AppendTo[rimTris, {base + 2, base + 4, base + 6}];

AppendTo[rimTris, {base + 4, base + 8, base + 6}];

AppendTo[rimTris, {base + 3, base + 7, base + 4}];

AppendTo[rimTris, {base + 7, base + 8, base + 4}];

AppendTo[rimTris, {base + 1, base + 2, base + 5}];

AppendTo[rimTris, {base + 2, base + 6, base + 5}];];

{rimVerts, rimTris}];

(*8.生成并合并*)

edgeWidth = 0.030; edgeSegs = 25;

sphereR = 1.0; shellThick = 0.06;

{pVerts, pTris} =

GenerateSolidPrismsSouth[allEdges, edgeWidth, edgeSegs, sphereR,

shellThick];

allZ = pVerts[[All, 3]];

zMax = Max[allZ];

zRimBottom = zMax - 0.015;

{rVerts, rTris} =

GenerateBowlRim[zRimBottom, 0.0, sphereR, shellThick, 100];

rimOffset = Length[pVerts];

rTrisOffset = Map[# + rimOffset &, rTris, {2}];

combinedVerts = Join[pVerts, rVerts];

combinedTris = Join[pTris, rTrisOffset];

combinedMesh =

Graphics3D[

GraphicsComplex[

combinedVerts, {EdgeForm[], GrayLevel[0.15], Polygon[pTris],

EdgeForm[], GrayLevel[0.65], Polygon[rTrisOffset]}],

Lighting -> {{"Ambient", GrayLevel[0.4]}, {"Directional",

White, {0, 0, 3}}, {"Directional", GrayLevel[0.25], {2, 1, -1}}},

Boxed -> False, SphericalRegion -> True, ViewPoint -> {2, 1, 1.8},

ViewVertical -> {0, 0, 1}, ImageSize -> 800,

PlotLabel -> Style["Hyperbolic Bowl Model", 16, Bold]];

combinedMesh

